
Глоссарий 

ФБ – функциональный блок. 

 

1 Цель документа 

 

Настоящее руководство представляет собой описание библиотеки CmpSysExec, которая позволяет 

организовать доступ к терминалу ОС Linux для вызова внешних утилит из приложения контроллера. 

Реализация библиотеки находится в Linux (библиотека CODESYS представляет собой только 

интерфейс), поэтому может меняться в зависимости от версии прошивки контроллера. В данном 

документе описана версия библиотеки 3.5.17.30. 

 

 

ПРИМЕЧАНИЕ 
Библиотека может использоваться только в контроллерах ОВЕН, программируемых в среде 
CODESYS V3.5. 

 

 

ПРИМЕЧАНИЕ 
Библиотека является внешней (external) и не может быть использована в симуляции или на 
виртуальном контроллере. 

 

 

ПРИМЕЧАНИЕ 
POU библиотеки не могут использоваться для вызова утилит, работающих в интерактивном 
режиме или создающих новый терминал: ssh, timeout и т. д. 

 

 

ПРИМЕЧАНИЕ 
POU библиотеки возвращают только строки из стандартного потока вывода (STDOUT). 
Получение строк из потока вывода ошибок (STDERR) не предусмотрено. 

2 Описание библиотеки CmpSysExec 

2.1 Установка библиотеки 

 

Библиотека CmpSysExec входит в состав пакета таргет-файлов ОВЕН, который доступен для загрузки 

на сайте  ОВЕН в разделе CODESYS V3/Сервисное ПО. Установка библиотеки происходит 

автоматически при установке пакета. Для установки пакета в CODESYS в меню Инструменты следует 

выбрать пункт Менеджер пакетов, после чего нажать Установить и указать путь к файлу библиотеки. 

 

2.2 Добавление библиотеки в проект CODESYS 

 

Для добавления библиотеки СmpSysExec в проект CODESYS в Менеджере библиотек следует 

нажать кнопку Добавить библиотеку и в строке поиска ввести CmpSysExec,  после чего выбрать из 

списка нужную библиотеку и нажать ОК. 

 
Рисунок 2.1 – Добавление библиотеки CmpSysExec 

 

После добавления библиотека появится в списке Менеджера библиотек: 

Рисунок 2.2 – Список библиотек проекта 

 

 

ПРИМЕЧАНИЕ 
При обращении к функциям и ФБ библиотеки следует перед их названием указывать префикс 
CmpSysExec (пример: CmpSysExec.SysExecute). 

  

http://www.owen.ru/


2 Описание библиотеки CmpSysExec 

 

5 

2.3 Описание библиотеки 

2.3.1 Функция SysExecute 

 

Функция SysExecute используется для отправки команды в терминал Linux. 

Функция выполняется в блокирующем режиме – то есть приостанавливает поток на время выполнения, 

в связи с чем работа приложения CODESYS на это время останавливается. 

Для запуска функции в фоновом режиме следует после текста команды добавить символ & (например, 

‘echo “test” &’). 

 

 

ПРИМЕЧАНИЕ 
Функция SysExecute должна вызываться исключительно по триггеру (но не в цикле программы). 

 

Таблица 2.1 – Описание входов и выходов функции SysExecute 

Имя 
переменной 

Тип Описание 

 
Входные переменные 

 

sys_command STRING Команда,  отправляемая в терминал Linux 

 

 

Ниже приведен пример использования функции. Когда переменная xEnable примет значение TRUE, в 

терминал будет отправлена команда /sbin/reboot, в результате чего контроллер перезагрузится. 

 

 

ПРИМЕЧАНИЕ 
Во время отладки функция SysExecute рекомендуется не создавать загрузочного приложения 
– в случае некорректного использования функции это может привести к негативным 
последствиям. Например, если в приведенном ниже коде переменная xEnable будет являться 
энергонезависимой (retain), то контроллер будет циклически перезагружаться. 

 

PROGRAM PLC_PRG 

VAR 

   xEnable: BOOL; 

   sCommand: STRING := '/sbin/reboot'; 

END_VAR 

 

// область редактора кода 

 

IF xEnable THEN 

  

   CmpSysExec.sysExecute(sCommand); 

   xEnable := FALSE; 

 

END_IF 

  

https://ru.wikipedia.org/wiki/%D0%9F%D0%BE%D1%82%D0%BE%D0%BA_%D0%B2%D1%8B%D0%BF%D0%BE%D0%BB%D0%BD%D0%B5%D0%BD%D0%B8%D1%8F

2.3.2 ФБ SysExecute2 

 

Функциональный блок SysExecute2 используется для отправки команды в терминал Linux и получения 

ответа. 

ФБ выполняется в неблокирующем режиме – то есть освобождает поток после подачи команды и через 

определенное время забирает ее ответ. 

 

 

ПРИМЕЧАНИЕ 

Количество одновременно (в пределах цикла) используемых экземпляров ФБ – не более 9. 
В случае использования более девяти экземпляров ФБ, работать будут только первые 9 (в 
порядке вызова). 

 

 

ПРИМЕЧАНИЕ 
ФБ может использоваться для выполнения команд, циклически возвращающих результат 
(например, top).  

 

Таблица 2.2 – Описание входов и выходов ФБ SysExecute2 

Имя 
переменной 

Тип Описание 

 
Входные переменные 

 

cmd STRING(255) Команда,  отправляемая в терминал Linux 

 
Выходные переменные 

 

output_string STRING(255) 

Текущая строка ответа. После отправки команды (это может занять 
несколько циклов) в этой переменной последовательно будут 
появляться  строки ответа. Так как обработка одной строки может 
занять несколько циклов, то в эту переменную могут быть записаны 
пустые строки, поэтому необходимо производить в программе 
соответствующую проверку (см. пример ниже) 

complete BOOL 

Флаг завершения работы блока. Принимает значение TRUE на 
следующий цикл после получения последней строки ответа 
(справедливо только для команд, возвращающих конечное 
количество строк ответа) 

 

 

Пример работы с ФБ приведен ниже.  

 

PROGRAM PLC_PRG 

VAR 

   xEnable:  BOOL; 

   sCommand:  STRING(c_usiMaxCommandLen) := 'ifconfig'; 

   fbSysExecute2: CmpSysExec.sysExecute2; 

   asOutput:  ARRAY [1..c_uiMaxOutputCount] OF STRING(c_usiMaxOutputLen); 

  

   i:   UINT; 

   fbEnableTrig:  R_TRIG; 

END_VAR 

VAR CONSTANT 

   c_uiMaxOutputCount: UINT  := 30; 

   c_usiMaxCommandLen: USINT := 255; 

   c_usiMaxOutputLen: USINT := 255; 

END_VAR 

 

// область редактора кода 

 

fbEnableTrig(CLK := xEnable); 

 

 

 

https://ru.wikipedia.org/wiki/%D0%9F%D0%BE%D1%82%D0%BE%D0%BA_%D0%B2%D1%8B%D0%BF%D0%BE%D0%BB%D0%BD%D0%B5%D0%BD%D0%B8%D1%8F


2 Описание библиотеки CmpSysExec 

 

7 

IF fbEnableTrig.Q THEN 

  

   i := 1; 

 

END_IF 

 

IF xEnable THEN 

  

   fbSysExecute2 

   ( 

      cmd     := sCommand,  

      output_string => asOutput[i] 

    ); 

 

    IF NOT(fbSysExecute2.complete) AND fbSysExecute2.output_string <> '' AND  

       i < c_uiMaxOutputCount THEN 

   

       i := i + 1; 

   

    END_IF 

 

    xEnable := NOT(fbSysExecute2.complete); 

 

END_IF 

 

Когда переменная xEnable принимает значение TRUE, то переменная i принимает значение 1 (позиция 

начальной строки массива для записи) и в терминал отправляется команда ifconfig которая, выводит 

информацию о сетевых интерфейсах контроллера. Полученный ответ построчно записывается в 

массив переменных типа STRING с названием asOutput. После завершения работы ФБ входу xEnable 

присваивается значение FALSE для прекращения вызова. В переменной i будет отображаться 

значение, соответствующее количеству строк вывода + 1. 

 

Значения констант могут быть подстроены под требования конкретного положения, но максимальная 

длина строк команды и ответа не может превышать 255 – это связано с ограничениями реализации ФБ. 

 

 
Рисунок 2.3 – Результат выполнения команды ifconfig через ФБ SysExecute2 

Рисунок 2.4 – Результат выполнения команды ifconfig в терминале Linux 

2.3.3 ФБ SysExecute3 

 

Функциональный блок SysExecute3 используется для отправки команды в терминал Linux и получения 

ответа. В отличие от ФБ SysExecute2, интерфейс данного блока соответствует CAA Behavior Model 

(PLCopen Behavior Model). 

ФБ выполняется в неблокирующем режиме – то есть освобождает поток после подачи команды и через 

определенное время забирает ее ответ. 

 

 

ПРИМЕЧАНИЕ 
Количество одновременно (в пределах цикла) используемых экземпляров ФБ – не более 9. 
В случае использования более девяти экземпляров ФБ, работать будут только первые 9 (в 
порядке вызова). 

 

 

ПРИМЕЧАНИЕ 
ФБ не может использоваться для выполнения команд, циклически возвращающих результат 
(например, top).  

 

 

Таблица 2.3 – Описание входов и выходов ФБ SysExecute3 

Имя 
переменной 

Тип Описание 

 
Входные переменные 

 

xExecute BOOL По переднему фронту происходит отправка команды 

sCommand STRING(255) Команда,  отправляемая в терминал Linux 

xAbort STRING 
По переднему фронту происходит прерывание работы блока (c 
помощью вызова функции pclose) 

 
Выходные переменные 

 

sOutput STRING(255) 

Текущая строка ответа. После отправки команды (это может занять 
несколько циклов) в этой переменной последовательно будут 
появляться  строки ответа. Так как обработка одной строки может 
занять несколько циклов, то в эту переменную могут быть записаны 
пустые строки, поэтому необходимо производить в программе 
соответствующую проверку (см. пример ниже). Последняя строка 
ответа, выводимая в момент xDone = TRUE, всегда является 
пустой 

xDone BOOL 
Флаг завершения работы блока. Принимает значение TRUE на 
следующий цикл после получения последней строки ответа 

xAborted BOOL 
Флаг прерывания работы блока (см. вход xAbort). Принимает 
значение TRUE после завершения выполняемого процесса 

xError BOOL Флаг ошибки 

xBusy BOOL Флаг «ФБ в процессе работы» 

eError UDINT 
Код ошибки. Возможные значения: 
1 – превышено число одновременно вызываемых экземпляров ФБ 

 

Пример работы с ФБ приведен ниже.  

 

 

 

 

 

 

 

 

 

https://ru.wikipedia.org/wiki/%D0%9F%D0%BE%D1%82%D0%BE%D0%BA_%D0%B2%D1%8B%D0%BF%D0%BE%D0%BB%D0%BD%D0%B5%D0%BD%D0%B8%D1%8F
https://www.opennet.ru/man.shtml?topic=pclose&category=3&russian=0

PROGRAM PLC_PRG 

VAR 

   xExecute:  BOOL; 

   sCommand:  STRING(c_usiMaxCommandLen) := 'ifconfig'; 

   fbSysExecute3: CmpSysExec.sysExecute3; 

   asOutput:  ARRAY [1..c_uiMaxOutputCount] OF STRING(c_usiMaxOutputLen); 

  

   i:          UINT; 

   fbEnableTrig:  R_TRIG; 

END_VAR 

VAR CONSTANT 

   c_uiMaxOutputCount: UINT  := 30; 

   c_usiMaxCommandLen: USINT := 255; 

   c_usiMaxOutputLen: USINT := 255; 

END_VAR 

 

// область редактора кода 

 

fbEnableTrig(CLK := xExecute); 

 

IF fbEnableTrig.Q THEN 

  

   i := 1; 

 

END_IF 

 

fbSysExecute3 

( 

   xExecute := xExecute,  

   sCommand := sCommand 

); 

 

IF fbSysExecute3.xBusy AND fbSysExecute3.sOutput <> '' AND i < c_uiMaxOutputCount THEN 

  

   asOutput[i] := fbSysExecute3.sOutput; 

   i := i + 1; 

 

END_IF 

 

IF fbSysExecute3.xDone THEN 

  

   xExecute := FALSE; 

 

END_IF 

 

 

Когда переменная xExecute принимает значение TRUE, то переменная i принимает значение 1 

(позиция начальной строки массива для записи) и в терминал отправляется команда ifconfig, которая 

выводит информацию о сетевых интерфейсах контроллера. Полученный ответ построчно 

записывается в массив переменных типа STRING с названием asOutput. В переменной i будет 

отображаться значение, соответствующее количеству строк вывода + 1. 

 

В рамках примера – после завершения работы ФБ (когда его выход xDone примет значение TRUE) 

переменной xExecute присваивается значение FALSE; в реальном приложении можно было бы 

сгенерировать в переменной xExecute единичный импульс для запуска ФБ; в этом случае его работа 

завершилась бы единичным импульсом на выходе xDone. Важно помнить, что запуск ФБ происходит 

по переднему фронту его входа xExecute – то есть этот вход перед запуском должен иметь значение 

FALSE. 

 

Значения констант могут быть подстроены под требования конкретного положения, но максимальная 

длина строк команды и ответа не может превышать 255 – это связано с ограничениями реализации ФБ. 

Рисунок 2.5 – Результат выполнения команды ifconfig через ФБ SysExecute3 

 

 
Рисунок 2.6 – Результат выполнения команды ifconfig в терминале Linux 

2.3.4 ФБ SysExecute4 

 

Функциональный блок SysExecute4 используется для отправки команды в терминал Linux и получения 

ответа. В отличие от ФБ SysExecute2 и SysExecute3, данный блок возвращает вывод команды не 

построчно, а в виде массива. 

ФБ выполняется в неблокирующем режиме – то есть освобождает поток после подачи команды и через 

определенное время забирает ее ответ. 

 

 

ПРИМЕЧАНИЕ 
Количество одновременно (в пределах цикла) используемых экземпляров ФБ – не более 9. 
В случае использования более девяти экземпляров ФБ, работать будут только первые 9 (в 
порядке вызова). 

 

 

ПРИМЕЧАНИЕ 
ФБ не может использоваться для выполнения команд, циклически возвращающих результат 
(например, top).  

 

Таблица 2.4 – Описание входов и выходов ФБ SysExecute4 

Имя 
переменной 

Тип Описание 

 
Входные переменные 

 

xExecute BOOL По переднему фронту происходит отправка команды 

sCommand STRING(255) Команда,  отправляемая в терминал Linux 

xAbort STRING 
По переднему фронту происходит прерывание работы блока (c 
помощью вызова функции pclose) 

 
Выходные переменные 

 

sOutput 
ARRAY [1..255 ] OF 

STRING(255) 

Текущая строка ответа. После отправки команды (это может 
занять несколько циклов) в этой переменной последовательно 
будут появляться  строки ответа. 

cntReadStr USINT Число строк вывода 

xOverFlow BOOL 
Флаг переполнения. Принимает значение TRUE, если число 
строк вывода команды превышает 255 

xDone BOOL 
Флаг завершения работы блока. Принимает значение TRUE 
после получения ответа 

xAborted BOOL 
Флаг прерывания работы блока (см. вход xAbort). Принимает 
значение TRUE после завершения выполняемого процесса 

xError BOOL Флаг ошибки 

xBusy BOOL Флаг «ФБ в процессе работы» 

eError UDINT 
Код ошибки. Возможные значения: 
1 – превышено число одновременно вызываемых экземпляров 
ФБ 

 

Пример работы с ФБ приведен ниже.  

 

 

 

 

 

 

 

 

 

 

 

https://ru.wikipedia.org/wiki/%D0%9F%D0%BE%D1%82%D0%BE%D0%BA_%D0%B2%D1%8B%D0%BF%D0%BE%D0%BB%D0%BD%D0%B5%D0%BD%D0%B8%D1%8F
https://www.opennet.ru/man.shtml?topic=pclose&category=3&russian=0

PROGRAM PLC_PRG 

VAR 

   xExecute:  BOOL; 

   sCommand:  STRING(c_usiMaxCommandLen) := 'ifconfig'; 

   fbSysExecute4: CmpSysExec.sysExecute4; 

   asOutput:  ARRAY [1..c_uiMaxOutputCount] OF STRING(c_usiMaxOutputLen); 

   uiOutputLineCount:    UINT; 

   xOutputArrayOverflow: BOOL; 

  

   i:   UINT; 

END_VAR 

VAR CONSTANT 

   c_uiMaxOutputCount: UINT  := 255; 

   c_usiMaxCommandLen: USINT := 255; 

   c_usiMaxOutputLen: USINT := 255; 

END_VAR 

 

// область редактора кода 

 

fbSysExecute4 

( 

   xExecute   := xExecute,  

   sCommand   := sCommand 

); 

 

IF fbSysExecute4.xDone THEN 

  

   FOR i := 1 TO MIN(fbSysExecute4.cntReadStr, c_uiMaxOutputCount) DO 

   

      asOutput[i] := fbSysExecute4.sOutput[i]; 

  

   END_FOR 

  

   uiOutputLineCount    := fbSysExecute4.cntReadStr; 

   xOutputArrayOverflow := fbSysExecute4.xOverFlow; 

 

   xExecute := FALSE; 

 

END_IF 

 

Когда переменная xExecute принимает значение TRUE, то в терминал отправляется команда ifconfig, 

которая выводит информацию о сетевых интерфейсах контроллера. Полученный ответ построчно 

записывается в массив переменных типа STRING с названием asOutput. В переменной 

uiOutputLineCount будет отображаться значение, соответствующее количеству строк вывода. 

 

В рамках примера – после завершения работы ФБ (когда его выход xDone примет значение TRUE) 

переменной xExecute присваивается значение FALSE; в реальном приложении можно было бы 

сгенерировать в переменной xExecute единичный импульс для запуска ФБ; в этом случае его работа 

завершилась бы единичным импульсом на выходе xDone. Важно помнить, что запуск ФБ происходит 

по переднему фронту его входа xExecute – то есть этот вход перед запуском должен иметь значение 

FALSE. 

 

Значения констант могут быть подстроены под требования конкретного положения, но максимальная 

длина строк команды и ответа не может превышать 255 – это связано с ограничениями реализации ФБ. 

Рисунок 2.7 – Результат выполнения команды ifconfig через ФБ SysExecute4 

 

 
Рисунок 2.8 – Результат выполнения команды ifconfig в терминале Linux 

2.3.5 ФБ SysExecute5 

 

Функциональный блок SysExecute5 используется для отправки команды в терминал Linux и получения 

ответа. В отличие от ФБ SysExecute3 выполняемая команда и ответ на нее представляют собой не 

строки, а указатели на строки. Длина команд и ответов ограничена 2047 символами. 

ФБ выполняется в неблокирующем режиме – то есть освобождает поток после подачи команды и через 

определенное время забирает ее ответ. 

 

 

ПРИМЕЧАНИЕ 
Количество одновременно (в пределах цикла) используемых экземпляров ФБ – не более 2. 
В случае использования более двух экземпляров ФБ, работать будут только первые 2 (в 
порядке вызова). 

 

 

ПРИМЕЧАНИЕ 
ФБ не может использоваться для выполнения команд, циклически возвращающих результат 
(например, top).  

 

 

Таблица 2.5 – Описание входов и выходов ФБ SysExecute5 

Имя 
переменной 

Тип Описание 

 
Входные переменные 

 

xExecute BOOL По переднему фронту происходит отправка команды 

xAbort BOOL 
По переднему фронту происходит прерывание работы блока 
(c помощью вызова функции pclose) 

pCommand POINTER TO BYTE Указатель на строку команды (не более 2047 символов) 

szCommand DWORD Длина строки команды 

pOutput POINTER TO BYTE 

Указатель на текущую строку ответа. После отправки 
команды (это может занять несколько циклов) в этой 
переменной последовательно будут появляться строки 
ответа. Так как обработка одной строки может занять 
несколько циклов, то в эту переменную могут быть записаны 
пустые строки, поэтому необходимо производить в 
программе соответствующую проверку (см. пример ниже). 
Последняя строка ответа, выводимая в момент xDone = 
TRUE, всегда является пустой. При вызове ФБ на этот вход 
обязательно должна быть передан адрес переменной (даже 
если ответ не нужен) 

szOutput DWORD 
Размер буфера строки ответа в байтах. При вызове ФБ на 
этот вход обязательно должна быть передана переменная 
(даже если ответ не нужен) 

 
Выходные переменные 

 

xDone BOOL 
Флаг завершения работы блока. Принимает значение TRUE 
на следующий цикл после получения последней строки 
ответа 

xAborted BOOL 
Флаг прерывания работы блока (см. вход xAbort). 
Принимает значение TRUE после завершения выполняемого 
процесса 

xError BOOL Флаг ошибки 

xBusy BOOL Флаг «ФБ в процессе работы» 

eError UDINT 
Код ошибки. Возможные значения: 
1 – превышено число одновременно вызываемых 
экземпляров ФБ 

uiOutputSize UINT Длина текущей строки ответа 

 

https://ru.wikipedia.org/wiki/%D0%9F%D0%BE%D1%82%D0%BE%D0%BA_%D0%B2%D1%8B%D0%BF%D0%BE%D0%BB%D0%BD%D0%B5%D0%BD%D0%B8%D1%8F
https://www.opennet.ru/man.shtml?topic=pclose&category=3&russian=0

Пример работы с ФБ приведен ниже. 

 

PROGRAM PLC_PRG 

VAR 

   xExecute:  BOOL; 

   sCommand:  STRING(c_uiMaxCommandLen) := 'ifconfig'; 

   fbSysExecute5: CmpSysExec.sysExecute5; 

   asOutput:  ARRAY [1..c_uiMaxOutputCount] OF STRING(c_uiMaxOutputLen); 

   sOutput:  STRING(c_uiMaxOutputLen); 

   uiOutputSize:  UINT; 

  

   i:   UINT; 

   fbEnableTrig:  R_TRIG; 

END_VAR 

VAR CONSTANT 

   c_uiMaxOutputCount: UINT  := 30; 

   c_uiMaxCommandLen: UINT  := 2047; 

   c_uiMaxOutputLen: UINT  := 2047; 

END_VAR 

 

// область редактора кода 

 

fbEnableTrig(CLK := xExecute); 

 

IF fbEnableTrig.Q THEN 

  

   i := 1; 

 

END_IF 

 

fbSysExecute5 

( 

   xExecute    := xExecute,  

   pCommand    := ADR(sCommand),  

   szCommand   := TO_DWORD(STU.StrLenA(ADR(sCommand) ) ), 

   pOutput     := ADR(sOutput), 

   szOutput    := SIZEOF(sOutput), 

   uiOutputSize   => uiOutputSize 

); 

 

IF fbSysExecute5.xBusy AND sOutput <> '' AND i < c_uiMaxOutputCount THEN 

  

   asOutput[i] := sOutput; 

   i := i + 1; 

 

END_IF 

 

IF fbSysExecute5.xDone THEN 

  

   xExecute := FALSE; 

 

END_IF 

 

Когда переменная xExecute принимает значение TRUE, то переменная i принимает значение 1 

(позиция начальной строки массива для записи) и в терминал отправляется команда ifconfig, которая 

выводит информацию о сетевых интерфейсах контроллера. Полученный ответ построчно 

записывается в массив переменных типа STRING с названием asOutput. В переменной i будет 

отображаться значение, соответствующее количеству строк вывода + 1. 

 

В рамках примера – после завершения работы ФБ (когда его выход xDone примет значение TRUE) 

переменной xExecute присваивается значение FALSE; в реальном приложении можно было бы 

сгенерировать в переменной xExecute единичный импульс для запуска ФБ; в этом случае его работа 

завершилась бы единичным импульсом на выходе xDone. Важно помнить, что запуск ФБ происходит 

по переднему фронту его входа xExecute – то есть этот вход перед запуском должен иметь значение 

FALSE. 

Значения констант могут быть подстроены под требования конкретного положения, но максимальная 

длина строк команды и ответа не может превышать 2047 – это связано с ограничениями реализации 

ФБ. 

 

 

ПРИМЕЧАНИЕ 
Для работы со строками, длина которых превышает 255 символов, в коде программы 
(конкатенация, определение длины и т. д.) следует использовать функции библиотеки 
StringUtils, поскольку функции библиотеки Standard не поддерживают обработку длинных 
строк. 
 

 
Рисунок 2.9 – Результат выполнения команды ifconfig через ФБ SysExecute5 

 

 
Рисунок 2.10 – Результат выполнения команды ifconfig в терминале Linux 

2.4 Таблица сравнения POU библиотеки 

 

Параметр SysExecute SysExecute2 SysExecute3 SysExecute4 SysExecute5 

Тип POU Функция ФБ 

Тип выполнения 
Синхронный 

(блокирующий) 
Асинхронный (неблокирующий) 

Макс. число 
одновременно 
вызываемых 
экземпляров 

- 9 9 9 2 

Макс. длина 
строки ответа 

- 255 255 255 2047 

Вывод Отсутствует 
Построчный, без ограничений на 

количество выводимых строк 

Массив, до 255 
строк, с 

возможностью 
переполнения 

Построчный, 
без 

ограничений на 
количество 
выводимых 

строк 

 

Принцип работы: 

• SysExecute – однократное выполнение команды, для которой не требуется получения вывода; 

• SysExecute2 – ФБ выполняется, пока происходит его вызов. Подходит для команд, циклически 

возвращающих результат (например, top); 

• SysExecute3, SysExecute4, SysExecute5 – ФБ запускается по переднему фронту на входе 

xExecute и построчно возвращает результат. Соответственно, число строк результата должно 

быть конечным (нельзя вызывать команды типа top). После завершения вывода на выходе 

xDone устанавливается значение TRUE. 

 


